Matrices, Geometry&*Mathematica*

Authors: Bruce Carpenter, Bill Davis and Jerry Uhl ©2001

Producer: Bruce Carpenter

Publisher: Math Everywhere, Inc.

MGM.09 Eigensense: Diagonalizable Matrices, Matrix Exponential, Matrix Powers and Dynamical Systems

TUTORIALS

T.1) Continuous dynamical systems

Systems of linear homogeneous differential equations {x'[t],y'[t]} = A.{x[t],y[t]} solved with the matrix exponential {x[t],y[t]} = e^{At}.starter

If you know how matrix exponentials plot out, you won't have much difficulty with this stuff.

T.1.a.i) Coefficient matrix of a system of linear homogeneous differential equations

Take this system of linear homogeneous differential equations

x'[t] = 0.3 x[t] − 0.6 y[t],
y'[t] = 0.1 x[t] + 0.8 y[t]

Come up with a matrix A (called the coefficient matrix) to write this system in matrix form

{x'[t], y'[t]} = A.{x[t], y[t]}..

Answer:

Look at the system of linear homogeneous differential equations

x'[t] = 0.3 x[t] − 0.6 y[t],

y'[t] = 0.1 x[t] + 0.8 y[t].

Now make this matrix:

```
A = {{0.3, -0.6}, {0.1, 0.8}};
MatrixForm[A]
```

$$\begin{pmatrix} 0.3 & -0.6 \\ 0.1 & 0.8 \end{pmatrix}$$

See what you get when you hit this matrix on {x[t], y[t]}:

```
Clear[x, y, t];
    A.{x[t], y[t]}
```

{0.3 x[t] - 0.6 y[t], 0.1 x[t] + 0.8 y[t]}

Because

x'[t] = 0.3 x[t] − 0.6 y[t],

y'[t] = 0.1 x[t] + 0.8 y[t] ,

This tells you that:

{x'[t], y'[t]} = A.{x[t], y[t]}.

Check:

```
ColumnForm[Thread[{x'[t], y'[t]} == A.{x[t], y[t]}]]
```

x′[t] == 0.3 x[t] - 0.6 y[t]
y′[t] == 0.1 x[t] + 0.8 y[t]

This checks and you're out of here.

T.1.a.ii) Slamming out a solution of a system of linear homogeneous differential equations {x'[t],y'[t]} = A.{x[t],y[t]} with the matrix exponential

Stay with the same setup as in part i)
Use the matrix exponential to slam out formulas {x[t], y[t]} for solutions of

x'[t] = 0.3 x[t] − 0.6 y[t],
y'[t] = 0.1 x[t] + 0.8 y[t]
x[0] = xstarter
y[0] = ystarter

in terms of xstarter and ystarter.

Answer:

The system is

x'[t] = 0.3 x[t] − 0.6 y[t],

y'[t] = 0.1 x[t] + 0.8 y[t]

The matrix form of this system is:

{x'[t], y'[t]} = A.{x[t], y[t]}

where the coefficient matrix A is:

```
A = (0.3  -0.6
     0.1   0.8);
MatrixForm[A]
```

$$\begin{pmatrix} 0.3 & -0.6 \\ 0.1 & 0.8 \end{pmatrix}$$

Getting formulas for solutions {x[t],y[t]} of

{x'[t], y'[t]} = A.{x[t], y[t]}

in terms of xstarter and ystarter , is really easy.

You just hit the matrix exponential E^{At} on {xstarter, ystarter}

and put

{x[t], y[t]} = E^{At}.{xstarter, ystarter}:

```
Clear[x, y, xstarter, ystarter];
{x[t_], y[t_]} = MatrixExp[A t].{xstarter, ystarter}
```

{(3. $e^{0.5t}$ - 2. $e^{0.6t}$) xstarter + (6. $e^{0.5t}$ - 6. $e^{0.6t}$) ystarter, (-1. $e^{0.5t}$ + 1. $e^{0.6t}$) xstarter + (-2. $e^{0.5t}$ + 3. $e^{0.6t}$) ystarter}

Check whether

{x′[t], y′[t]} = A.{x[t], y[t]}:

```
Expand[{x'[t], y'[t]}]
```

{1.5 $e^{0.5t}$ xstarter - 1.2 $e^{0.6t}$ xstarter + 3. $e^{0.5t}$ ystarter - 3.6 $e^{0.6t}$ ystarter, -0.5 $e^{0.5t}$ xstarter + 0.6 $e^{0.6t}$ xstarter - 1. $e^{0.5t}$ ystarter + 1.8 $e^{0.6t}$ ystarter}

```
Expand[A.{x[t], y[t]}]
```

{1.5 $e^{0.5t}$ xstarter - 1.2 $e^{0.6t}$ xstarter + 3. $e^{0.5t}$ ystarter - 3.6 $e^{0.6t}$ ystarter, -0.5 $e^{0.5t}$ xstarter + 0.6 $e^{0.6t}$ xstarter - 1. $e^{0.5t}$ ystarter + 1.8 $e^{0.6t}$ ystarter}

Good.

Check whether

{x[0], y[0]} = {xstarter, ystarter}

```
{x[0], y[0]}
```

{1. xstarter, 1. ystarter}

Everything checks.

T.1.a.iii) Why that worked

Explain why that worked.
And while you're at it explain this:
When you go with a 2D diagonalizable coefficient matrix A and put

{x′[t], y′[t]} = A.{x[t], y[t]},

then you can slam out formulas for the solution {x[t], y[t]} that starts at a given point

{x[0], y[0]} = {a, b}

by putting

{x[t], y[t]} = E^{At}.{a, b}.

Answer:

Just check it:

Step 1: Check whether {x′[t], y′[t]} = A.{x[t], y[t]}:

Take the formula

{x[t], y[t]} = E^{At}.{a, b}

and differentiate with respect to t and see what you get:

{x′[t], y′[t]} = D[{x[t], y[t]}, t]
= D[E^{At}.{a, b}, t]
= A.E^{At}.{a, b}

The fact that D[E^{At}, t] = A.E^{At} is discussed in a Basic)

So {x[t], y[t]} = E^{At}.{a, b} gives

{x′[t], y′[t]} = A.E^{At}.{a, b}

and this is the same as

{x′[t], y′[t]} = A.{x[t], y[t]}

because

{x[t], y[t]} = E^{At}.{a, b}

Step 2: Check whether {x[0], y[0]} = {a, b}:

Take the formula

{x[t], y[t]} = E^{At}.{a, b}

and plug in t = 0 and see what you get:

{x[0], y[0]} = E^{A0}.{a, b} = IdentityMatrix.{a, b} = {a, b}

The result: Everything checks.

T.1.b.i) Using eigenvalues of the coefficient matrix to predict how solutions plot out

Here's a 2D system of homogeneous linear differential equations:

```
Clear[x, y, t];
linearsystem =
  {x′[t], y′[t]} == {-0.8 x[t] + 0.1 y[t], 0.9 x[t] - 1.75 y[t]};

ColumnForm[Thread[linearsystem]]
```

x′[t] == -0.8 x[t] + 0.1 y[t]
y′[t] == 0.9 x[t] - 1.75 y[t]

The coefficient matrix of this system is:

```
A = (-0.8   0.1
      0.9  -1.75);
MatrixForm[A]
```

$\begin{pmatrix} -0.8 & 0.1 \\ 0.9 & -1.75 \end{pmatrix}$

The eigenvalues of A are:

```
Eigenvalues[A]
```

{-1.83681, -0.713195}

Use this information to predict how solutions {x[t],y[t]} of this linear system plot out as t gets large.
Illustrate your answer with a good plot.

Answer:

Given a starting point {a, b}, the solution {x[t], y[t]} with {x[0], y[0]} = {a, b} is

$\{x[t], y[t]\} = E^{A\,t}.\{a, b\}$.

Take another look at the eigenvalues of A::

```
Eigenvalues[A]
```

{-1.83681, -0.713195}

Both negative.

According to the Basics, this tells you that no matter what {a, b} is, you are guaranteed that

$\{x[t], y[t]\} = E^{A\,t}.\{a, b\} \to \{0, 0\}$

as t gets large..

The upshot: As t gets large, all solutions of this linear system are attracted to {0,0}.

See it happen for four solutions starting at random points:

```
Clear[trajectoryplot, eigenplot, scaler,
  eigenvector, pointer, starter, thigh, k, j];

trajectoryplot[thigh_, starter_] :=
  ParametricPlot[MatrixExp[A t].starter, {t, 0, thigh},
   PlotStyle -> {{Thickness[0.012], CadmiumOrange}},
   DisplayFunction -> Identity];


pointer[thigh_, starter_] :=
  ArrowHead[MatrixExp[A thigh].starter,
   (D[MatrixExp[A t], t] /. t -> thigh).starter,
               HeadSize -> 0.6, VectorColor -> Black,
   Aperture -> 0.4];

starterplots = Table[
   Graphics[{Navy, PointSize[0.03], Point[starter[j]]}], {j, 1, 4}];

starter[1] = {Random[Real, {4, 10}], Random[Real, {4, 10}]};
```

```
starter[2] = {Random[Real, {-10, -4}], Random[Real, {-10, -4}]};
starter[3] = {Random[Real, {-10, -4}], Random[Real, {4, 10}]};
starter[4] = {Random[Real, {4, 10}], Random[Real, {-10, -4}]};

Clear[solutionplots, thigh];
solutionplots[thigh_] :=
 Show[Table[trajectoryplot[thigh, starter[k]], {k, 1, 4}] ,
    Table[pointer[thigh, starter[k]], {k, 1, 4}], starterplots,
  PlotRange -> All, PlotLabel -> "Solutions for 0< t" < thigh,
  DisplayFunction -> Identity];
Show[solutionplots[0.2], DisplayFunction -> $DisplayFunction];
Show[solutionplots[0.5], DisplayFunction -> $DisplayFunction];
Show[solutionplots[2.0], DisplayFunction -> $DisplayFunction];
Show[solutionplots[4.0], DisplayFunction -> $DisplayFunction];
Show[solutionplots[10.0], DisplayFunction -> $DisplayFunction];
Show[solutionplots[50.0], DisplayFunction -> $DisplayFunction];
```













Grab and animate.

Sucked down the drain at {0,0}.

And you knew this in advance because both eigenvalues of the coefficient matrix A are **negative.**

T.1.b.ii) Another linear system with complex eigenvalues

Here's a new 2D system of homogeneous linear differential equations:

```
Clear[x, y, t];
linearsystem =
  {x'[t], y'[t]} == {0.8 x[t] + 1.7 y[t], -2.9 x[t] - 1.75 y[t]};

ColumnForm[Thread[linearsystem]]
```

x'[t] == 0.8 x[t] + 1.7 y[t]
y'[t] == -2.9 x[t] - 1.75 y[t]

The coefficient matrix of this system is:

```
A = ( 0.8    1.7
     -2.9  -1.75 );
MatrixForm[A]
```

$\begin{pmatrix} 0.8 & 1.7 \\ -2.9 & -1.75 \end{pmatrix}$

The eigenvalues of A are:

```
Eigenvalues[A]
```

{-0.475 + 1.81779 i, -0.475 - 1.81779 i}

Use this information to predict how solutions {x[t],y[t]} of this linear system plot out as t gets large.
Illustrate your answer with a good plot.

Answer:

Given a starting point {a, b}, the solution {x[t], y[t]} with {x[0], y[0]} = {a, b} is

$\{x[t], y[t]\} = E^{A\,t}.\{a, b\}$.

Take another look at the eigenvalues of A::

```
Eigenvalues[A]
```

{-0.475 + 1.81779 i, -0.475 - 1.81779 i}

This are of the form $p + I\,q$ and $p - I\,q$ with $p < 0$ and $q \neq 0$.

Because $p < 0$ this tells you that no matter what {a, b} is, you are guaranteed that

$\{x[t], y[t]\} = E^{A\,t}.\{a, b\} \to \{0, 0\}$

as t gets large..

Because $q \neq 0$ this tells you that no matter what {a, b} is, you are guaranteed that

$\{x[t], y[t]\} = E^{A\,t}.\{a, b\}$ **swirls** its way into {0,0}

as t advances..

The upshot: As t gets large, all solutions of this linear system **swirl** their way to {0,0}.

See it happen for four solutions starting at random points:

```
Clear[trajectoryplot, eigenplot, scaler,
  pointer, eigenvector, starter, thigh, k, j];

trajectoryplot[thigh_, starter_] :=
  ParametricPlot[MatrixExp[A t].starter, {t, 0, thigh},
   PlotStyle -> {{Thickness[0.012], CadmiumOrange}},
   DisplayFunction -> Identity];


pointer[thigh_, starter_] :=
```

```
  ArrowHead[MatrixExp[A thigh] .starter,
    (D[MatrixExp[A t], t] /. t -> thigh) .starter,
               HeadSize -> 1, VectorColor -> Black, Aperture -> 0.4];

starterplots = Table[
   Graphics[{Navy, PointSize[0.03], Point[starter[j]]}], {j, 1, 4}];

starter[1] = {Random[Real, {4, 10}], Random[Real, {4, 10}]};
starter[2] = {Random[Real, {-10, -4}], Random[Real, {-10, -4}]};
starter[3] = {Random[Real, {-10, -4}], Random[Real, {4, 10}]};
starter[4] = {Random[Real, {4, 10}], Random[Real, {-10, -4}]};

Clear[solutionplots, thigh];
solutionplots[thigh_] :=
 Show[Table[trajectoryplot[thigh, starter[k]], {k, 1, 4}] ,
    Table[pointer[thigh, starter[k]], {k, 1, 4}], starterplots,
  PlotRange -> All, PlotLabel -> "Solutions for 0< t" < thigh,
  DisplayFunction -> Identity];
ranger = 13;
Show[solutionplots[1.0],
  PlotRange -> {{-ranger, ranger}, {-ranger, ranger}},
  DisplayFunction -> $DisplayFunction];


Show[solutionplots[2.0],
  PlotRange -> {{-ranger, ranger}, {-ranger, ranger}},
  DisplayFunction -> $DisplayFunction];

Show[solutionplots[7.0],
  PlotRange -> {{-ranger, ranger}, {-ranger, ranger}},
  DisplayFunction -> $DisplayFunction];
```







Grab and animate.

Swirling down the drain at {0,0}.

And you knew this in advance because the eigenvalues of the coefficient matrix A are of the form p + I q and p – I q with p < 0 and q ≠ 0.

T.2) Discrete dynamical systems: Customer dynamics solved with matrix powers

Thanks Jacob Knoop,Sales Manager of Southern Ag Equipment, for help on this problem.

□ T.2.a.i) Setting up a the discrete difference equations

Southern Ag Equipment, importers of specialized carrot harvesters, has been through a lengthy process of computerizing their sales records through the Macintosh software QuickBook Pro from Intuit.

Their customer base consists of the 1500 active carrot farmers in the US.
Through computer analysis, they have noticed:
- 71% of their customers who order in year k also order in year k + 1
- 45% of their customers who did not order in year k do order in year k + 1.

Put
x[k] = fraction of their customers who do order in year k
y[k] = fraction of their customers who do not order in year k.

Explain why this gives
$$x[k+1] = 0.71\, x[k] + 0.45\, y[k]$$
and
$$y[k+1] = 0.29\, x[k] + 0.55\, y[k].$$

□ Answer:

The relationship
$$x[k+1] = 0.71\, x[k] + 0.45\, y[k]$$
makes sense because 71% of their customers who ordered in year k also order in year k+1

and

45% of their customers who did not order in year k do order in year k + 1.

The relationship
$$y[k+1] = 0.29\, x[k] + 0.55\, y[k].$$
makes sense because 29% of their customers who ordered in year k also do not order in year k+1 and

55% of their customers who did not order in year k do not order in year k + 1.

□ T.2.a.ii) Solving the discrete difference equations with matrix powers

Stay with everything just as it was in part i) above.
Use the relationships
$x[k+1] = 0.71\, x[k] + 0.45\, y[k]$ and
$y[k+1] = 0.29\, x[k] + 0.55\, y[k]$

to explain why
$$\{x[k], y[k]\} = .\begin{pmatrix} 0.71 & 0.45 \\ 0.29 & 0.55 \end{pmatrix}^k .\{x[0], y[0]\}$$

□ Answer:

The relationships
$x[k+1] = 0.71\, x[k] + 0.45\, y[k]$ and
$y[k+1] = 0.29\, x[k] + 0.55\, y[k]$
are the same as
$$\{x[k+1], y[k+1]\} = \begin{pmatrix} 0.71 & 0.45 \\ 0.29 & 0.55 \end{pmatrix}.\{x[k], y[k]\}.$$

To see why just multiply it out:

```
Clear[x, y, k];
(0.71  0.45
 0.29  0.55).{x[k], y[k]}
```

```
{0.71 x[k] + 0.45 y[k], 0.29 x[k] + 0.55 y[k]}
```

This gives:
$$\{x[1], y[1]\} = \begin{pmatrix} 0.71 & 0.45 \\ 0.29 & 0.55 \end{pmatrix}.\{x[0], y[0]\}.$$

$$\{x[2], y[2]\} = \begin{pmatrix} 0.71 & 0.45 \\ 0.29 & 0.55 \end{pmatrix}.\{x[1], y[1]\}$$

$$= \begin{pmatrix} 0.71 & 0.45 \\ 0.29 & 0.55 \end{pmatrix}.\begin{pmatrix} 0.71 & 0.45 \\ 0.29 & 0.55 \end{pmatrix}.\{x[0], y[0]\}$$

$$= \begin{pmatrix} 0.71 & 0.45 \\ 0.29 & 0.55 \end{pmatrix}^2 .\{x[0], y[0]\}.$$

$$\{x[3], y[3]\} = \begin{pmatrix} 0.71 & 0.45 \\ 0.29 & 0.55 \end{pmatrix}.\{x[2], y[2]\}$$

$$= \begin{pmatrix} 0.71 & 0.45 \\ 0.29 & 0.55 \end{pmatrix}.\begin{pmatrix} 0.71 & 0.45 \\ 0.29 & 0.55 \end{pmatrix}^2 .\{x[0], y[0]\}$$

$$= \begin{pmatrix} 0.71 & 0.45 \\ 0.29 & 0.55 \end{pmatrix}^3 .\{x[0], y[0]\}$$

This leads to the general pattern:

$$\{x[k], y[k]\} = \begin{pmatrix} 0.71 & 0.45 \\ 0.29 & 0.55 \end{pmatrix}^k .\{x[0], y[0]\}.$$

□ T.2.a.iii) The ultimate ratio

Stay with the same set up as in part i) and remember that
x[k] = fraction of their customers who order in year k
y[k] = fraction of their customers who do not order in year k.
Predict the ultimate ratio $\frac{x[k]}{y[k]}$ as k gets large and interpret the result.

□ Answer:

You know
$\{x[k], y[k]\} = A^k.\{x[0], y[0]\}$.
with
$A = \begin{pmatrix} 0.71 & 0.45 \\ 0.29 & 0.55 \end{pmatrix}$.

Enter A and check its eigenvalues and eigenvectors:

```
A = (0.71  0.45
     0.29  0.55);
Clear[eigenvalue];
{eigenvalue[1], eigenvalue[2]} = Eigenvalues[A]
{1., 0.26}
```

Note that | eigenvalue[1] | > | eigenvalue[2] |.

Check the eigenvectors:

```
Clear[eigenvector];
{eigenvector[1], eigenvector[2]} = Eigenvectors[A]
{{0.840571, 0.541701}, {-0.707107, 0.707107}}
```

Remember from the Basics:
If | eigenvalue[1] | > | eigenvalue[2] |, then the ultimate direction of $A^k.X$ as k gets large is in the direction of eigenvector[1] (unless X is a multiple of eigenvector[2]).
Here
X = {x[0], y[0]}
cannot be a multiple of eigenvector[2] because {x[0], y[0]} cannot have negative entries.
So the ultimate ratio x[k]/y[k] as k gets large is approximately:

```
UltimateRatio = 0.840571 / 0.541701
1.55173
```

In the long run,Southern Ag will have about 1.55 customers who order for each customer who doesn't.

□ T.2.a.iv) Limiting values

Keep everything the same as in the parts above. Give the limiting values of x[k] and y[k] and interpret the result.

□ Answer:

The point to remember here is that
x[k] = fraction of their customers who order in year k
y[k] = fraction of their customers who do not order in year k.
So
x[k] + y[k] = 1.
You also know that as k gets large $\frac{x[k]}{y[k]}$ closes in on:

```
UltimateRatio
1.55173
```

So in the long run,
x[k] ≈ UltimateRatio y[k].
Plug this into
x[k] + y[k] = 1
to get
(UltimateRatio y[k]) + y[k] ≈ 1.
This is the same as
(UltimateRatio + 1) y[k] ≈ 1.
This tells you that in the long run (for k large), y[k] is approximately:

```
ylongrun = 1 / (1 + UltimateRatio)
0.391892
```

And this tells you that in the long run (for k large), x[k] is approximately:

```
xlongrun = 1 - ylongrun
0.608108
```

The interpretation:
In the long run, about 61% of Southern Ag's customers will order in a given year and about 39% will not.

See it happen by starting with x[0] = 0 and y[0] = 1, the situation when Southern Ag opened for business:

```
A = {{0.71, 0.45}, {0.29, 0.55}};
starter = {0, 1};
Clear[x, y, k];
ColumnForm[
 Table[{{x[k], y[k]}, "=", MatrixPower[A, k].{0, 1}}, {k, 0, 10}]]
{{x[0], y[0]}, =, {0, 1}}
{{x[1], y[1]}, =, {0.45, 0.55}}
{{x[2], y[2]}, =, {0.567, 0.433}}
{{x[3], y[3]}, =, {0.59742, 0.40258}}
{{x[4], y[4]}, =, {0.605329, 0.394671}}
{{x[5], y[5]}, =, {0.607386, 0.392614}}
{{x[6], y[6]}, =, {0.60792, 0.39208}}
{{x[7], y[7]}, =, {0.608059, 0.391941}}
{{x[8], y[8]}, =, {0.608095, 0.391905}}
{{x[9], y[9]}, =, {0.608105, 0.391895}}
{{x[10], y[10]}, =, {0.608107, 0.391893}}
```

Compare:

```
{xlongrun, ylongrun}
{0.608108, 0.391892}
```

The model predicts that after about 4 years, Southern Ag's long term market penetration will be approximately achieved.

T.3) Discrete dynamical systems: Linear homogeneous difference equations solved with matrix powers

If you know how matrix powers plot out, you won't have much difficulty with this stuff.

□ T.3.a.i) Linear difference equation in matrix form

Take the linear homogeneous difference equation
y[k + 2] − 0.5 y[k + 1] + 0.27 y[k] = 0,
Come up with a matrix A (called the coefficient matrix) to write this difference equation in matrix form
$\{y[k], y[k+1]\} = A^k.\{y[0], y[1]\}$
for all k = 0, 1, 2, 3, 4, 5, ...

□ Answer:

Here $A^2 = A.A$, $A^3 = A.A.A$, $A^4 = A.A.A.A$ .etc.

Look at the difference equation:
y[k + 2] = 0.5 y[k + 1] − 0.27 y[k];
Now make this matrix:

```
A = (0    1
     0.5  -0.27);
MatrixForm[A]
(0    1
 0.5  -0.27)
```

See what you get when you hit this matrix on {y[k], y[k + 1]}:

```
Clear[y, k];
A.{y[k], y[k + 1]}
{y[1 + k], 0.5 y[k] - 0.27 y[1 + k]}
```

Because
y[k + 2] = 0.5 y[k + 1] − 0.27 y[k],
This tells you that:
{y[k + 1], y[k + 2]} = A.{y[k], y[k + 1]}
for all k's.

Plug in k = 0 to get
{y[1], y[2]} = A.{y[0], y[1]}.

Plug in k = 1 into
{y[k + 1], y[k + 2]} = A.{y[k], y[k + 1]}
to get
{y[2], y[3]} = A.{y[1], y[2]} = A.A.{y[0], y[1]}
Reason: {y[1],y[2]}= A.{y[0],y[1]}

Plug in k = 2 into
{y[k + 1], y[k + 2]} = A.{y[k], y[k + 1]}
to get
{y[3], y[4]} = A.{y[2], y[3]} = A.A.A.{y[0], y[1]}
Reason: {y[2],y[3]} = A.A.{y[0],y[1]}

Continuing on in this way, you are led to the matrix form {y[k], y[k + 1]} = A^k.{y[0], y[1]}

of the given difference equation.

□T.3.a.ii) Using the matrix to slamming out y[2], y[3], y[4], y[5], y[6], y[7], y[8], y[9] and y[10] in terms of y[0] and y[1]

Stay with the same setup as in part i)
Use the matrix form of the difference equation to slam out
y[2], y[3], y[4], y[5], y[6], y[7], y[8], y[9] and y[10]
in terms of y[0] and y[1].

□Answer:

The difference equation is

y[k + 2] = 0.5 y[k + 1] – 0.27 y[k];

The matrix form is:

{y[k], y[k + 1]} = A^k.{y[0], y[1]}where A is:

```
MatrixForm[A]
```

$$\begin{pmatrix} 0 & 1 \\ 0.5 & -0.27 \end{pmatrix}$$

Use the matrix form

{y[k], y[k + 1]} = A^k.{y[0], y[1]}

with k = 1 to see that

{y[1], y[2]}

is given by:

```
k = 1;
Clear[y];
MatrixPower[A, k].{y[0], y[1]}
```

{1. y[1], 0.5 y[0] – 0.27 y[1]}

Use the matrix form

{y[k], y[k + 1]} = A^k.{y[0], y[1]}

with k = 2 to see that

{y[2], y[3]}

is given by:

```
k = 2;
MatrixPower[A, k].{y[0], y[1]}
```

{0.5 y[0] – 0.27 y[1], –0.135 y[0] + 0.5729 y[1]}

Here comes {y[3], y[4]} in terms of y[0] and y[1]:

```
k = 3;
MatrixPower[A, k].{y[0], y[1]}
```

{–0.135 y[0] + 0.5729 y[1], 0.28645 y[0] – 0.289683 y[1]}

Here comes {y[4], y[5]} in terms of y[0] and y[1]:

```
k = 4;
MatrixPower[A, k].{y[0], y[1]}
```

{0.28645 y[0] – 0.289683 y[1], –0.144842 y[0] + 0.364664 y[1]}

Here comes {y[5], y[6]} in terms of y[0] and y[1]:

```
k = 5;
MatrixPower[A, k].{y[0], y[1]}
```

{–0.144842 y[0] + 0.364664 y[1], 0.182332 y[0] – 0.243301 y[1]}

Here comes {y[6], y[7]} in terms of y[0] and y[1]:

```
k = 6;
MatrixPower[A, k].{y[0], y[1]}
```

{0.182332 y[0] – 0.243301 y[1], –0.12165 y[0] + 0.248023 y[1]}

Here comes {y[7], y[8]} in terms of y[0] and y[1]:

```
k = 7;
MatrixPower[A, k].{y[0], y[1]}
```

{–0.12165 y[0] + 0.248023 y[1], 0.124012 y[0] – 0.188617 y[1]}

Here comes {y[8], y[9]} in terms of y[0] and y[1]:

```
k = 8;
MatrixPower[A, k].{y[0], y[1]}
```

{0.124012 y[0] – 0.188617 y[1], –0.0943084 y[0] + 0.174938 y[1]}

Here comes {y[9], y[10]} in terms of y[0] and y[1]:

```
k = 9;
MatrixPower[A, k].{y[0], y[1]}
```

{–0.0943084 y[0] + 0.174938 y[1], 0.0874691 y[0] – 0.141542 y[1]}

If you want a list, you can get it

```
Clear[k];
    ColumnForm[
 Table[{y[k] , (MatrixPower[A, k].{y[0], y[1]})[[1]]}, {k, 2, 20}]]
```

{y[2], 0.5 y[0] – 0.27 y[1]}
{y[3], –0.135 y[0] + 0.5729 y[1]}
{y[4], 0.28645 y[0] – 0.289683 y[1]}
{y[5], –0.144842 y[0] + 0.364664 y[1]}
{y[6], 0.182332 y[0] – 0.243301 y[1]}
{y[7], –0.12165 y[0] + 0.248023 y[1]}
{y[8], 0.124012 y[0] – 0.188617 y[1]}
{y[9], –0.0943084 y[0] + 0.174938 y[1]}
{y[10], 0.0874691 y[0] – 0.141542 y[1]}
{y[11], –0.0707709 y[0] + 0.125685 y[1]}
{y[12], 0.0628427 y[0] – 0.104706 y[1]}
{y[13], –0.052353 y[0] + 0.0911133 y[1]}
{y[14], 0.0455566 y[0] – 0.0769535 y[1]}
{y[15], –0.0384768 y[0] + 0.0663341 y[1]}
{y[16], 0.0331671 y[0] – 0.056387 y[1]}
{y[17], –0.0281935 y[0] + 0.0483915 y[1]}
{y[18], 0.0241958 y[0] – 0.0412592 y[1]}
{y[19], –0.0206296 y[0] + 0.0353358 y[1]}
{y[20], 0.0176679 y[0] – 0.0301703 y[1]}

□T.3.a.iii) Using eigenvalues to size up the situation

Stay with the same setup as in part i) and look at the coefficient matrix A for the difference equation y[k + 2] – 0.5 y[k + 1] + 0.27 y[k] = 0:

```
MatrixForm[A]
```

$$\begin{pmatrix} 0 & 1 \\ 0.5 & -0.27 \end{pmatrix}$$

Check the eigenvalues of A:

```
Eigenvalues[A]
```

{–0.854878, 0.584878}

Use this eigenvalue information to determine what happens to y[k] as k gets large.

□Answer:

Look at the eigenvalues again:

```
Eigenvalues[A]
```

{–0.854878, 0.584878}

The absolute values of both eigenvalues are less than 1.

One of the Basics says :

If A is any diagonalizable matrix and the
absolute values of the of all the eigenvalues of A are less than 1,
then no matter what X is, you are guaranteed that

A^k X → {0, 0}.

So no matter what {y[0], y[1]} are, you are guaranteed that

{y[k], y[k + 1]} = A^k.{y[0], y[1]} → {0, 0} as k gets large.

The upshot :

y[k] –> 0 as k gets large.

See it happen :

```
Clear[k];
    ColumnForm[Table[
  {y[k] , (MatrixPower[A, k].{y[0], y[1]})[[1]]}, {k, 10, 100, 10}]]
```

{y[10], 0.0874691 y[0] – 0.141542 y[1]}
{y[20], 0.0176679 y[0] – 0.0301703 y[1]}
{y[30], 0.00368057 y[0] – 0.0062927 y[1]}
{y[40], 0.000767276 y[0] – 0.00131185 y[1]}
{y[50], 0.000159954 y[0] – 0.000273483 y[1]}
{y[60], 0.0000333456 y[0] – 0.0000570129 y[1]}
{y[70], 6.95157×10^{-6} y[0] – 0.0000118855 y[1]}
{y[80], 1.44919×10^{-6} y[0] – 2.47777×10^{-6} y[1]}
{y[90], 3.02114×10^{-7} y[0] – 5.16541×10^{-7} y[1]}
{y[100], 6.29816×10^{-8} y[0] – 1.07683×10^{-7} y[1]}

Withering away to 0.

□T.3.b.i) What is a linear homogeneous difference equation?

What is a linear homogeneous difference equation?

□Answer:

A linear homogeneous difference equation is any difference equation that can be put in matrix form as above.

□T.3.b.ii) The message

What's the message?

□Answer:

Once you have put a linear homogeneous difference equation in matrix form, you can analyze its behavior in by knowing how the matrix power function plots out. This is what was done above. It you want to see some more, go on.

□T.3.c.i) Another one

Use the matrix form of

y[k + 2] – 1.4 y[k + 1] + 0.3 y[k] = 0

to study the behavior of solutions y[k] as k->infinity

□Answer:

Look at the difference equation:

y[k + 2] = 1.4 y[k + 1] – 0.3 y[k];

Make the coefficient matrix:

```
A = {{0, 1}, {1.4, -0.3}};
MatrixForm[A]
```

$$\begin{pmatrix} 0 & 1 \\ 1.4 & -0.3 \end{pmatrix}$$

The matrix form of

$y[k + 2] - 1.4\, y[k + 1] + 0.3\, y[k] = 0$

is

$\{y[k], y[k + 1]\} = A^k.\{y[0], y[1]\}$.

Check the eigenvalues of A

```
Clear[eigval, eigvect];
    {eigval[1], eigval[2]} = Eigenvalues[A]
```

{-1.34269, 1.04269}

Now check the eigenvectors:

```
{eigvect[1], eigvect[2]} = Eigenvectors[A]
```

{{-0.597315, 0.802007}, {0.692179, 0.721726}}

The juicy info:

| eigval[1] | > 1 and

| eigval[1] | > | eigval[2] |

This tells you that unless {y[0], y[1]} is a multiple of eigenvector[2], you are guaranteed that as k gets large,

|| {y[k], y[k + 1]} || = (|| A)^k.{y[0], y[1]} || –> infinity.

And this tells that unless {y[0], y[1]} is a multiple of eigvect[2], you are guaranteed that as k gets large,

{y[k], y[k + 1]}

gravitates to the line through {0,0} defined by eigvect[1]; so that as k gets large

$\frac{y[k+1]}{y[k]} \to \frac{0.782245}{-0.622971}$:

```
-0.782245/0.622971
```

-1.25567

On the other hand, because | eigval[2] | < 1:

```
eigval[2]
```

1.04269

This tells you that if {y[0], y[1]} is a multiple of eigvect[2], you are guaranteed that as k gets large,

|| {y[k], y[k + 1]} || = (|| A)^k.{y[0], y[1]} || –> {0, 0}.

And this tells you that k gets large,

{y[k], y[k + 1]}

gravitates to the line through {0,0} defined by eigvect2].:

```
eigvect[2]
```

{0.692179, 0.721726}

This tells you that as k gets large

$\frac{y[k+1]}{y[k]} \to \frac{0.6909}{0.72295}$:

```
0.6909/0.72295
```

0.955668

T.3.c.ii) Yet another one

Use the matrix form of

$y[k + 2] - 1.2\, y[k + 1] + 0.3\, y[k] = 0$

to study the behavior of solutions y[k] as k->infinity

Answer:

Look at the difference equation:

$y[k + 2] = 1.2\, y[k + 1] - 0.3\, y[k]$;

Make the coefficient matrix:

```
A = {{0, 1}, {1.2, -0.3}};
MatrixForm[A]
```

$$\begin{pmatrix} 0 & 1 \\ 1.2 & -0.3 \end{pmatrix}$$

The matrix form of

$y[k + 2] - 1.4\, y[k + 1] + 0.3\, y[k] = 0$

is

$\{y[k], y[k + 1]\} = A^k.\{y[0], y[1]\}$.

Check the eigenvalues of A

```
Clear[eigval, eigvect];
    {eigval[1], eigval[2]} = Eigenvalues[A]
```

{-1.25567, 0.955667}

Now check the eigenvectors:

```
{eigvect[1], eigvect[2]} = Eigenvectors[A]
```

{{-0.622971, 0.782245}, {0.72295, 0.6909}}

The juicy info:

| eigval[1] | > 1 and

| eigval[1] | > | eigval[2] |

This tells you that unless {y[0], y[1]} is a multiple of eigenvector[2], you are guaranteed that as k gets large,

|| {y[k], y[k + 1]} || = (|| A)^k.{y[0], y[1]} || –> infinity.

And this tells that unless {y[0], y[1]} is a multiple of eigvect[2], you are guaranteed that as k gets large,

{y[k], y[k + 1]}

gravitates to the line through {0,0} defined by eigvect[1]; so that as k gets large

$\frac{y[k+1]}{y[k]} \to \frac{0.782245}{-0.622971}$:

```
-0.782245/0.622971
```

-1.25567

On the other hand, because | eigval[2] | < 1:

```
eigval[2]
```

0.955667

This tells you that if {y[0], y[1]} is a multiple of eigvect[2], you are guaranteed that as k gets large,

|| {y[k], y[k + 1]} || = (|| A)^k.{y[0], y[1]} || –> {0, 0}.

And this tells you that k gets large,

{y[k], y[k + 1]}

gravitates to the line through {0,0} defined by eigvect2].:

```
eigvect[2]
```

{0.72295, 0.6909}

This tells you that as k gets large

$\frac{y[k+1]}{y[k]} \to \frac{0.6909}{0.72295}$:

```
0.6909/0.72295
```

0.955668

T.4) How do you pronounce the words "eigenvector" and "eigenvalue"? What about the algebra behind calculating them?

T.4.a) How do you pronounce the words "eigenvector" and "eigenvalue"?

How do you pronounce the words "eigenvector" and "eigenvalue"?

Answer:

The words "eigenvector" and "eigenvalue" are bastardized words. "Eigen" comes from German, "vector" and "value" come from English. You pronounce "eigen" as if it were German and you pronounce "vector" and "value" as usual.

"Eigen" is pronounced I - gen with a hard g as in "garden" or "geezer," not the soft g as in "general" or "genetics."

The German word "eigen" corresponds to the English word "own."

T.4.b.i) Using roots of the characteristic polynomial Det[A - λ Identity] to calculate eigenvalues of a given square matrix A

Go with this sample matrix A:

```
dim = 4;
A = Table[Random[Real, {-5, 5}], {i, 1, dim}, {j, 1, dim}];
MatrixForm[A]
```

$$\begin{pmatrix} 1.30264 & -4.59659 & -2.5002 & 0.824036 \\ -2.70334 & -2.84353 & -2.92323 & -4.26716 \\ -2.97003 & -0.642021 & -1.09444 & -0.822921 \\ -3.20145 & 4.31296 & 4.79637 & 2.40866 \end{pmatrix}$$

Put characteristicpoly[λ] = Det[A - λ IdentityMatrix]:

```
Clear[characteristicpoly, λ];
characteristicpoly[λ_] = Det[A - λ IdentityMatrix[dim]]
```

$127.946 + 144.971\,\lambda - 5.10469\,\lambda^2 + 0.226673\,\lambda^3 + \lambda^4$

Find when characteristicpoly[λ] is 0:

```
Solve[characteristicpoly[λ] == 0, λ]
```

{{λ → -5.37322}, {λ → -0.859329},
 {λ → 3.00294 - 4.32345 i}, {λ → 3.00294 + 4.32345 i}}

Compare:

```
Eigenvalues[A]
```

{-5.37322, 3.00294 + 4.32345 i, 3.00294 - 4.32345 i, -0.859329}

Bingo.

The solutions λ of

characteristicpoly[λ] = 0

are the eigenvalues of A.

Explain why the same thing will work for any square matrix A.

Answer:

Saying that λ is an eigenvalue for a given matrix A is the same as saying that there is a vector X ≠ {0,0,. . . ,0} with

A.X = λ X.

This is the same as saying that there is a vector X ≠ {0,0,. . . ,0} with

A.X - λ X. = {0,0,. . . ,0}.

And this is the same as saying that there is a vector X ≠ {0,0,. . . ,0} with

(A - λ Identity).X = {0,0,. . . ,0}.

.

This is the same as saying that the matrix (A - λ Identity) hits a non-zero vector into {0,0,...,0}.

This is the same as saying that the matrix (A - λ Identity) has at least one zero SVD stretch factor.

This is the same as saying that

characteristicpoly[λ] = Det[(A - λ Identity)] = 0.

Try it out on more random square matrices A:

```
dim = Random[Integer, {2, 10}];
A = Table[Random[Real, {-5, 5}], {i, 1, dim}, {j, 1, dim}];
MatrixForm[A]
```

-1.65275	-0.399736	-1.28946	2.0586	-3.79721	-1.52361
-4.84979	4.41415	-0.0998541	-1.92703	2.65041	-1.40988
-2.39651	-4.08349	0.573645	-2.14272	-4.42648	1.55853
-3.33191	3.6802	3.77496	2.24557	-3.12828	-3.72847
0.427711	-2.3547	3.16118	-0.787069	-0.775076	4.16892
3.01097	-0.201223	4.32478	1.09594	-4.63944	-3.79134

```
Clear[characteristicpoly, λ];
characteristicpoly[λ_] = Det[A - λ IdentityMatrix[dim]]
```

$-3761.1 + 6208.44\,\lambda + 577.339\,\lambda^2 + 187.227\,\lambda^3 + 36.2228\,\lambda^4 - 1.0142\,\lambda^5 + \lambda^6$

```
Solve[characteristicpoly[λ] == 0, λ]
```

{{λ → -5.08734}, {λ → -0.767873 - 5.15365 i}, {λ → -0.767873 + 5.15365 i}, {λ → 0.56947}, {λ → 3.53391 - 5.94383 i}, {λ → 3.53391 + 5.94383 i}}

```
Eigenvalues[A]
```

{3.53391 + 5.94383 i, 3.53391 - 5.94383 i, -0.767873 + 5.15365 i, -0.767873 - 5.15365 i, -5.08734, 0.56947}

Rerun all four cells until you are satisfied.

T.4.b.ii) Using roots of the characteristic polynomial Det[A - λ Identity] to calculate eigenvectors of a given square matrix A

Go with this sample matrix A:

```
dim = 4;
A = Table[Random[Real, {-5, 5}], {i, 1, dim}, {j, 1, dim}];
MatrixForm[A]
```

1.72129	0.179435	-0.21309	3.35138
1.14777	3.62091	-1.88118	4.67119
2.37281	-3.62466	-3.7529	3.39965
-3.0549	3.73004	-1.91408	-0.813278

Put characteristicpoly[λ] = Det[A - λ IdentityMatrix]:

```
Clear[characteristicpoly, λ];
characteristicpoly[λ_] = Det[A - λ IdentityMatrix[dim]]
```

$-72.6722 - 60.2423\,\lambda - 22.3059\,\lambda^2 - 0.776022\,\lambda^3 + \lambda^4$

The roots of the characteristic polynomial are the eigenvalues of A:

```
Clear[eigenvalue, k];
eigenvalue[k_] := Eigenvalues[A][[k]];
Table[eigenvalue[k], {k, 1, dim}]
```

{6.22058, -2.80892, -1.31782 + 1.55642 i, -1.31782 - 1.55642 i}

How can you use the calculated eigenvalues to come up with the corresponding eigenvectors?

Answer:

If you have your hands on eigenvalue[k], then the corresponding eigenvectors X are solutions of

A.X = eigenvalue[k] X.

This is the same as saying that

A.X - eigenvalue[k] X = {0,0,.. . ,0}.

This is the same as saying that

(A - eigenvalue[k] Identity).X = {0,0,. . .,0}.

This is the same as saying that

X is in the **null space**

of the matrix (A - eigenvalue[k] Identity)

.

Here is the unit eigenvector corresponding to eigenvalue[1]:

```
eigenvector1 =
 Flatten[NullSpace[A - eigenvalue[1] IdentityMatrix[dim]]]
```

{0.309149, 0.871687, -0.120289, 0.360719}

Check to be sure that A.eigenvector1 = eigenvalue[1] eigenvector1:

```
A.eigenvector1
```

{1.92309, 5.4224, -0.748267, 2.24388}

```
eigenvalue[1] eigenvector1
```

{1.92309, 5.4224, -0.748267, 2.24388}

Here are the unit eigenvectors corresponding to each of the eigenvalues:

```
Clear[eigenvector, k];
eigenvector[k_] :=
 Flatten[NullSpace[A - eigenvalue[k] IdentityMatrix[dim]]];
ColumnForm[Table[eigenvector[k], {k, 1, dim}]]
```

{0.309149, 0.871687, -0.120289, 0.360719}
{-0.272333, -0.00618778, 0.864023, 0.423392}
{-0.524637, -0.256769 - 0.0170477 i, 0.24602 - 0.516618 i, 0.505143 - 0.27
{-0.524637, -0.256769 + 0.0170477 i, 0.24602 + 0.516618 i, 0.505143 + 0.27

Check them out:

```
Table[A.eigenvector[k] == eigenvalue[k] eigenvector[k], {k, 1, dim}]
```

{True, False, True, True}

Try it for more random square matrices A:

```
dim = 6;
A = Table[Random[Real, {-5, 5}], {i, 1, dim}, {j, 1, dim}];
MatrixForm[A]
```

2.72017	4.56112	0.0749542	4.38794	3.3954	-1.53482
-0.285601	3.17929	-3.32589	3.28574	4.92749	4.8279
0.526334	4.66484	1.80867	-4.84328	3.15353	3.2895
0.561562	-3.24293	1.20843	4.55946	-2.52436	2.57034
3.48826	4.99834	2.40069	3.1824	-4.90714	1.53316
-2.31371	-4.99689	3.41875	3.24741	-2.2412	-4.82479

```
Clear[eigenvalue, k];
eigenvalue[k_] := Eigenvalues[A][[k]];

Clear[eigenvector, k];
eigenvector[k_] :=
 Flatten[NullSpace[A - eigenvalue[k] IdentityMatrix[dim]]];
ColumnForm[Table[eigenvector[k], {k, 1, dim}]]
```

{-0.526985, -0.207715 - 0.432215 i, -0.264357 + 0.276495 i, 0.315573 - 0.2
{-0.526985, -0.207715 + 0.432215 i, -0.264357 - 0.276495 i, 0.315573 + 0.2
{0.34405, 0.175458, -0.132807, -0.216891, -0.754118, 0.466338}
{-0.419447, -0.00448829, 0.207137, 0.262895, 0.564166, -0.627488}
{-0.861576, 0.11552 - 0.0427601 i, -0.27477 - 0.165463 i, -0.0152197 - 0.0
{-0.861576, 0.11552 + 0.0427601 i, -0.27477 + 0.165463 i, -0.0152197 + 0.0

```
ColumnForm[Eigenvectors[A]]
```

{0.526985, 0.207715 + 0.432215 i, 0.264357 - 0.276495 i, -0.315573 + 0.249
{0.526985, 0.207715 - 0.432215 i, 0.264357 + 0.276495 i, -0.315573 - 0.249
{0.34405, 0.175458, -0.132807, -0.216891, -0.754118, 0.466338}
{0.419447, 0.00448829, -0.207137, -0.262895, -0.564166, 0.627488}
{0.861576, -0.11552 + 0.0427601 i, 0.27477 + 0.165463 i, 0.0152197 + 0.011
{0.861576, -0.11552 - 0.0427601 i, 0.27477 - 0.165463 i, 0.0152197 - 0.011

T.4.b.iii) Is the route of characteristic polynomials and null spaces a recommended way of calculating eigenvalues and eigenvectors?

Is the route of characteristic polynomials and null spaces a recommended way of calculating eigenvalues and eigenvectors in daily life?

Answer:

Heavens to Betsy - NO!

All good computer based mathematics processors such as *Mathematica* or MatLab have professionally written built-in instructions that are optimized for calculation of eigenvalues and eigenvectors.
Using these instructions is the recommended way of calculating eigenvalues and eigenvectors in daily life.

The route of characteristic polynomials and null spaces is the path of necessity for those unfortunate folks condemned to a life of pencil and paper hand calculation.

You should know about route of characteristic polynomials and null spaces because you might meet one of these unfortunate folks out on the street or in a coffee house sometime.

Take pity on them. These folks are forced to spend hour after hour in pure useless calculation.

□T.4.b.iv) Should you worry about calculation of eigenvectors and eigenvalues by hand?

Should you worry about calculation of eigenvectors and eigenvalues by hand?

□Answer:

Only if you enjoy self-inflicted pain.
In 2D,the cheat sheet chart problem following this one gives as easy a hand path as there is.

T.5) Traces, determinants and the 2D matrix exponential cheat sheet chart

□T.5.a.i) Eigenvalues of a 2D matrix A are

$$\frac{1}{2}\left(\text{trace} + \sqrt{\text{trace}^2 - 4\,\text{det}}\right) \text{ and } \frac{1}{2}\left(\text{trace} - \sqrt{\text{trace}^2 - 4\,\text{det}}\right)$$

Here comes a cleared matrix:

```
Clear[a, b, c, d];
A = (a  b
     c  d);
MatrixForm[A]
```

$$\begin{pmatrix} a & b \\ c & d \end{pmatrix}$$

Now calculate the eigenvalues of A:

```
Eigenvalues[A]
```

$$\left\{\frac{1}{2}\left(a + d - \sqrt{a^2 + 4\,b\,c - 2\,a\,d + d^2}\right), \frac{1}{2}\left(a + d + \sqrt{a^2 + 4\,b\,c - 2\,a\,d + d^2}\right)\right\}$$

Two quantities stand out:
$(a + d)$ and $a\,d - b\,c$
Folks have special names for these quantities. They say
trace of A $= a + d$
and
determinant of A $= a\,d - b\,c$.

```
{trace, det} = {a + d, Det[A]}
```

$\{a + d, -b\,c + a\,d\}$

You can easily express the eigenvalues of A in terms of the trace of A and the determinant of A:

```
ExpandAll[Eigenvalues[A]] ==
 ExpandAll[{1/2 (trace - Sqrt[trace^2 - 4 det]), 1/2 (trace + Sqrt[trace^2 - 4 det])}]
```

True

So the two eigenvalues of A are given by the formulas:
$\frac{1}{2}\left(\text{trace} + \sqrt{\text{trace}^2 - 4\,\text{det}}\right)$
and
$\frac{1}{2}\left(\text{trace} - \sqrt{\text{trace}^2 - 4\,\text{det}}\right)$.
These formulas make it possible for you to calculate the eigenvalues of a coefficient matrix with pencil and paper or with a cheap pocket calculator.

Lots of reference books have charts, of the type you will see below, that help you to see how the matrix exponential
$E^{A\,t}$.starter
plots out once you know the trace and the determinant.
In this problem, you will participate in building one of these charts. To start building the chart, you go with trace and determinant axes and then you plot the curve
$\text{trace}^2 - 4\,\text{determinant} = 0$:

```
cutoff =
  Plot[trace^2/4, {trace, -5, 5}, PlotStyle → {{Red, Thickness[0.01]}},
   AxesLabel → {"trace", "det"},
   PlotRange → {-6, 6}, DisplayFunction → Identity];
cutofflabel = Graphics[Text[FontForm[
     "\!\(trace\^2\) - 4 det == 0", {"Times", 10}], {3.0, -4.0}]];

pointer =
  Arrow[{2, 1} - {3.0, -3.6}, Tail → {3.0, -3.6}, VectorColor → Black];
```

```
Show[cutoff, cutofflabel,
  pointer, DisplayFunction → $DisplayFunction];
```



And then you annotate this plot as follows:

```
twosuckercutoff =
  Graphics[{Blue, Thickness[0.01], Line[{{-5, 0}, {0, 0}}]}];
twosuckerlabel = Graphics[Text[FontForm[
      " Both real and \n both negative", {"Times", 10}], {-4, 1}]];

twopropcutoff =
  Graphics[{Blue, Thickness[0.01], Line[{{0, 0}, {5, 0}}]}];
twoproplabel = Graphics[Text[FontForm[
      " Both real and \n both positive", {"Times", 10}], {4, 1}]];

propswirlcutoff = Graphics[
   {CadmiumOrange, Thickness[0.01], Line[{{0, 0}, {0, 6}}]}];
propswirllabel = Graphics[Text[
    FontForm[" p + I q \n and p - I q\n with p > 0 \n and q ≠ 0",
     {"Times", 10}], {2, 4}]];

pureswirlerlabel =
  Graphics[Text[FontForm[" p\n u\n r ", {"Times", 12}], {0, 7}]];

suckswirlcutoff = Graphics[
   {CadmiumOrange, Thickness[0.01], Line[{{0, 0}, {0, 6}}]}];
suckswirllabel = Graphics[Text[
    FontForm["p + I q \n and p - I q\n with p < 0 \n and q ≠ 0",
     {"Times", 10}], {-2, 4}]];

negdetcutoff =
  Graphics[{Blue, Thickness[0.01], Line[{{-5, 0}, {5, 0}}]}];
negdetlabel = Graphics[
   Text[FontForm["Both real and \n one > 0 and one < 0",
     {"Times", 10}], {0, -1.5}]];

pureswirlerlabel = Graphics[
   Text[FontForm["0 + I q and 0 - I q on positive vertical axis",
```

```
     {"Times", 10}], {0, 7}]];

chart = Show[cutoff, cutofflabel, pointer,
   twosuckercutoff, twosuckerlabel, twopropcutoff,
   twoproplabel, propswirlcutoff, propswirllabel,
   suckswirlcutoff, suckswirllabel, negdetcutoff, negdetlabel,
   pureswirlerlabel, DisplayFunction → $DisplayFunction];
```



The chart is not to be memorized.
It is to be called up and used when you decide you want it.

Some folks even carry a copy of this chart in their wallets.
How do folks use this chart to predict how $E^{A\,t}$.starter plots out?

□Answer:

Here it is for a sample matrix A

```
a = -1.4;
b = 0.9;
c = 0.4;
d = -0.5;
A = (a  b
     c  d);

MatrixForm[A]
```

$$\begin{pmatrix} -1.4 & 0.9 \\ 0.4 & -0.5 \end{pmatrix}$$

You calculate
trace $= a + d$
and
det $= a\,d - b\,c$
by hand. And then you plot the resulting point
{trace, det}
on the chart:

```
trace = a + d;
det = a d - b c;
chartpointpoint =
  Graphics[{Magenta, PointSize[0.04], Point[{trace, det}]}];

Show[chart, chartpointpoint];
```



Both eigenvalues are negative. This tells you that no matter what {a,b} is, you can be certain that

$E^{A t}$.starter → {0, 0} as t gets large.

Another:

```
a = -1.4;
b = -3.0;
c = 2.1;
d = 1.4;
A = (a b
     c d);
MatrixForm[A]

trace = a + d;
det = a d - b c;
chartpointpoint =
  Graphics[{HotPink, PointSize[0.04], Point[{trace, det}]}];

Show[chart, chartpointpoint];
```

$$\begin{pmatrix} -1.4 & -3. \\ 2.1 & 1.4 \end{pmatrix}$$



This point {trace,det}:

```
{trace, det}
```

{0, 4.34}

lands on the positive vertical axis.

This tells you that as t advances, $E^{A t}$.starter oscillates on an ellipse centered at {0,0}.

Here is the same thing for some random matrices:

```
a = Random[Real, {-2, 2}];
b = Random[Real, {-2, 2}];
c = Random[Real, {-2, 2}];
d = Random[Real, {-2, 2}];
A = (a b
     c d);
MatrixForm[A]

trace = a + d;
det = a d - b c;
chartpointpoint =
  Graphics[{HotPink, PointSize[0.04], Point[{trace, det}]}];

Show[chart, chartpointpoint];
```

$$\begin{pmatrix} -0.843032 & 1.43303 \\ 0.380053 & -1.9926 \end{pmatrix}$$



Rerun a few times.

Neat cheat sheet.

Print one up and put it in your wallet.

□T.5.a.ii) Explanations

Explain why the chart is correct.

□Answer:

Select the ones you want.
Reading them all could get tiresome.

□p + I q and p - I q with p < 0 and q ≠ 0 explanation

Here are some sample points {trace, det} that plot out in the two swirling sucker region of the chart:

```
Clear[trace, k];
    trace[k_] := trace[k] = Random[Real, {-4, -0.1}];
samplepoints = Table[{trace[k], (trace[k]^2) / 4 +
     Random[Real, {0, 5 - (trace[k]^2) / 4}]}, {k, 1, 100}];
samplepointplot = ListPlot[samplepoints, PlotStyle ->
    {Banana, PointSize[0.03]}, DisplayFunction -> Identity];
Show[samplepointplot, chart, PlotRange -> All,
  AxesLabel -> {"trace", "det"},
          DisplayFunction -> $DisplayFunction];
```



The p + I q and p - I q with p < 0 and q ≠ 0 region corresponds to the case that

trace of A < 0,

determinant of A > 0, and

$(\text{trace of A})^2 < 4$ determinant of A.

Explanation:

The eigenvalues of A are

$$p + I\,q = \frac{1}{2}\left(\text{trace} + \sqrt{\text{trace}^2 - 4\,\text{det}}\right)$$

and

$$p - I\,q = \frac{1}{2}\left(\text{trace} - \sqrt{\text{trace}^2 - 4\,\text{det}}\right).$$

Because

$\text{det} > 0$

and

$\text{trace}^2 - 4\,\text{det} < 0,$

you see q is not 0. This gives you the whirl.

Because trace < 0, you see that p < 0. This gives you the suck.

□0 + I q and 0 - I q explanation:

Here are some sample points {trace, det} that plot out in 0 + I q and 0 - I q region of the chart:

```
samplepoints = Table[{0, Random[Real, {0, 6}]}, {k, 1, 10}];
samplepointplot = ListPlot[samplepoints,
    PlotStyle -> {CadmiumOrange, PointSize[0.04]},
    DisplayFunction -> Identity];
Show[samplepointplot, chart, PlotRange -> All,
  AxesLabel -> {"trace", "det"},
          DisplayFunction -> $DisplayFunction];
```



The 0 + I q and 0 - I q region corresponds to the case that

trace of A = 0,

and

determinant of A > 0.

Explanation:

The eigenvalues of A are

$$p + I\,q = \frac{1}{2}\left(\text{trace} + \sqrt{\text{trace}^2 - 4\,\text{det}}\right)$$

and

$$p - I\,q = \frac{1}{2}\left(\text{trace} - \sqrt{\text{trace}^2 - 4\,\text{det}}\right).$$

Because

$\text{trace} = 0,$

$p + I\,q = \frac{1}{2}\left(0 + \sqrt{0 - 4\,\text{det}}\right)$

and

$p - I\,q = \frac{1}{2}\left(0 - \sqrt{0 - 4\,\text{det}}\right).$

And because det > 0,

$p + I\,q = \frac{1}{2}\left(0 + 2\,I\,\sqrt{\text{Abs}[\text{det}]}\right)$

and

$p - I\,q = \frac{1}{2}\left(0 - 2\,I\,\sqrt{\text{Abs}[\text{det}]}\right)$

□ p + I q and p - I q with p > 0 and q ≠ 0 explanation:

Here are some sample points {trace, det} that plot out in the p + I q and p - I q with p > 0 and q ≠ 0 region of the chart:

```
Clear[trace, k];
    trace[k_] := trace[k] = Random[Real, {0.1, 4}];
samplepoints = Table[ {trace[k],
     (trace[k]^2) / 4 + Random[Real, {0, 5 - (trace[k]^2) / 4}]},
    {k, 1, 100}];
samplepointplot =
  ListPlot[samplepoints, PlotStyle -> {Yellow, PointSize[0.03]},
   DisplayFunction -> Identity];
Show[samplepointplot, chart, PlotRange -> All,
  AxesLabel -> {"trace", "det"},
          DisplayFunction -> $DisplayFunction];
```



The two p + I q and p - I q with p > 0 and q ≠ 0 region signal the case that

trace of A > 0,

determinant of A > 0, and

(trace of A)^2 < 4 determinant of A.

Explanation:

The eigenvalues of A are

$p + I\,q = \frac{1}{2}\left(\text{trace} + \sqrt{\text{trace}^2 - 4\,\text{det}}\right)$

and

$p - I\,q = \frac{1}{2}\left(\text{trace} - \sqrt{\text{trace}^2 - 4\,\text{det}}\right).$

Because

$\text{trace}^2 - 4\,\text{det} < 0,$

you see q is not 0.

Because trace > 0, you see that p > 0.

□ Both real and both negative explanation:

Here are some sample points {trace, det} that plot out in the both real and both negative region of the chart:

```
Clear[trace, k];
    trace[k_] := trace[k] = Random[Real, {-5, -0.3}];
samplepoints = Table[
   {trace[k], Random[Real, {0, (trace[k]^2) / 4}]}, {k, 1, 100}];
samplepointplot = ListPlot[samplepoints, PlotStyle ->
    {ManganeseBlue, PointSize[0.03]}, DisplayFunction -> Identity];
Show[samplepointplot, chart, PlotRange -> All,
          DisplayFunction -> $DisplayFunction];
```



The both real and both negative region corresponds to the case that

trace of A < 0,

determinant of A > 0, and

(trace of A)^2 > 4 determinant of A.

Explanation:

The eigenvalues of A are

$\frac{1}{2}\left(\text{trace} - \sqrt{\text{trace}^2 - 4\,\text{det}}\right)$

and

$\frac{1}{2}\left(\text{trace} + \sqrt{\text{trace}^2 - 4\,\text{det}}\right).$

Because

$\text{trace}^2 - 4\,\text{det} > 0,$

you see that neither eigenvalue involves $I = \sqrt{-1}$. So there is no whirl.

Because det > 0, you are guaranteed that

$0 < \sqrt{\text{trace}^2 - 4\,\text{det}} < \sqrt{\text{trace}^2} = -\text{trace}.$

Remember trace < 0 in this case.

This guarantees that both eigenvalues

$\frac{1}{2}\left(\text{trace} - \sqrt{\text{trace}^2 - 4\,\text{det}}\right) < 0$

$\frac{1}{2}\left(\text{trace} + \sqrt{\text{trace}^2 - 4\,\text{det}}\right) < 0.$

The upshot: Both eigenvalues are negative.

□ Both real and both positive explanation:

Here are some sample points {trace, det} that plot out in the two pure propeller region of the chart:

```
Clear[trace, k];
    trace[k_] := trace[k] = Random[Real, {0.3, 5}];
samplepoints = Table[
   {trace[k], Random[Real, {0, (trace[k]^2) / 4}]}, {k, 1, 100}];
samplepointplot = ListPlot[samplepoints, PlotStyle ->
    {LightCoral, PointSize[0.03]}, DisplayFunction -> Identity];
Show[samplepointplot, chart, PlotRange -> All,
          DisplayFunction -> $DisplayFunction];
```



The both real and both positive region corresponds to case that

trace of A > 0,

determinant of A > 0, and

(trace of A)^2 > 4 determinant of A.

Explanation:

The eigenvalues of A are

$\frac{1}{2}\left(\text{trace} - \sqrt{\text{trace}^2 - 4\,\text{det}}\right)$

and

$\frac{1}{2}\left(\text{trace} + \sqrt{\text{trace}^2 - 4\,\text{det}}\right)$

Because

$\text{trace}^2 - 4\,\text{det} > 0,$

you see that neither eigenvalue involves $I = \sqrt{-1}$. So there is no whirl.

Because det > 0, you are guaranteed that

$0 < \sqrt{\text{trace}^2 - 4\,\text{det}} < \sqrt{\text{trace}^2} = \text{trace}.$

Remember trace > 0 in this case.

This guarantees that both eigenvalues

$\frac{1}{2}\left(\text{trace} - \sqrt{\text{trace}^2 - 4\,\text{det}}\right) > 0$

$\frac{1}{2}\left(\text{trace} + \sqrt{\text{trace}^2 - 4\,\text{det}}\right) > 0.$

The upshot: Both eigenvalues are positive.

▫Both real with one positive and one negative explanation

Here are some sample points {trace, det} that plot out in the both real with one positive and one negative region of the chart:

```
Clear[trace, k];
    trace[k_] := trace[k] = Random[Real, {-5, 5}];
samplepoints =
  Table[{trace[k], Random[Real, {-4, 0}]}, {k, 1, 100}];
samplepointplot = ListPlot[samplepoints, PlotStyle ->
    {Turquoise, PointSize[0.03]}, DisplayFunction -> Identity];
Show[samplepointplot, chart, PlotRange -> All,
          DisplayFunction -> $DisplayFunction];
```



The both real with one positive and one negative region corresponds to the case that determinant of A < 0.

Explanation:

The eigenvalues of A are

$\frac{1}{2}\left(\text{trace} - \sqrt{\text{trace}^2 - 4\,\text{det}}\right)$

and

$\frac{1}{2}\left(\text{trace} + \sqrt{\text{trace}^2 - 4\,\text{det}}\right)$.

Because det < 0, you are guaranteed that

$\text{trace}^2 - 4\,\text{det} > 0$,

so that neither eigenvalue involves $I = \sqrt{-1}$. So there is no whirl.

Also because det < 0, you are guaranteed that

$\sqrt{\text{trace}^2 - 4\,\text{det}} > \sqrt{\text{trace}^2} = \text{Abs[trace]}$.

This tells you that A has one negative eigenvalue - namely

$\frac{1}{2}\left(\text{trace} - \sqrt{\text{trace}^2 - 4\,\text{det}}\right)$.

And the other eigenvalue

$\frac{1}{2}\left(\text{trace} + \sqrt{\text{trace}^2 - 4\,\text{det}}\right)$

is automatically positive.